Complete the timing diagram for the BRAM defined on the opposing page.

-------------------------------------------------------------------------
-- Name:	Prof Jeff Falkinburg
-- Date:	Jan 29, 2017
[bookmark: _GoBack]-- File:	lec13_dp.vhdl
------------------------------------------------------------------------- 
library IEEE;		
use IEEE.std_logic_1164.all; 
use IEEE.NUMERIC_STD.ALL;			-- contains the unsigned data type 
library UNIMACRO;				-- This contains links to the Xilinx block RAM
use UNIMACRO.vcomponents.all;

entity lec12Dual_dp is
	Port(	clk: in  STD_LOGIC;
		n_reset : in  STD_LOGIC;
		cw: std_logic_vector(5 downto 0));
end lec12Dual_dp;

architecture behavior of lec12Dual_dp is
	signal writeInput, readOutput: std_logic_vector(17 downto 0);
	signal addrWrite, addrRead: unsigned(9 downto 0);
	signal vecAddrWrite, vecAddrRead : std_logic_vector(9 downto 0);
	signal reset: std_logic;
begin		
	process(clk)
	begin
		if (rising_edge(clk)) then
			if (n_reset = '0') then
				addrWrite <= (others => '0');
			elsif (cw(1 downto 0) = "01") then
				addrWrite <= addrWrite + 1;
			elsif (cw(1 downto 0) = "11") then
				addrWrite <= (others => '0');
			end if;
		end if;
	end process;

	addrRead <= addrWrite - 1;				-- Have the read follow the writes
	writeInput <= "10101010101010" & vecAddrWrite(3 downto 0);		
	reset <= not n_reset;				-- BRAM reset is active high
	vecAddrWrite <= std_logic_vector(addrWrite);	-- type conversion 
	vecAddrRead <= std_logic_vector(addrRead);	

	sampleMemory: BRAM_SDP_MACRO
		generic map (
			BRAM_SIZE => "18Kb", 	-- Target BRAM, "18Kb" or "36Kb"
			DEVICE => "7SERIES", -- Target device: "VIRTEX5", "VIRTEX6", "SPARTAN6, 7SERIES"
			DO_REG => 0, 			-- Optional output register disabled
			INIT => X"000000000000000000",	-- Initial values on output port
			INIT_FILE => "NONE",	-- Not sure how to initialize the RAM from a file
			WRITE_WIDTH => 18, 	-- Valid values are 1-72 (37-72 valid when BRAM_SIZE="36Kb")
			READ_WIDTH => 18, 	-- Valid values are 1-72 (37-72 valid when BRAM_SIZE="36Kb")
			SIM_COLLISION_CHECK => "NONE", 	-- Simulation collision check
			SRVAL => X"000000000000000000")	-- Set/Reset value for port output
		port map (
			DO => readOutput,			-- Output read data port width=READ_WIDTH 
			RDADDR => vecAddrRead,		-- Read address, width=FNC(BRAM_SIZE)
			RDCLK => clk,	 			-- 1-bit input clock
			RST => reset,			-- active high reset
			RDEN => cw(5),			-- read enable 
			REGCE => '1',				-- read output register enable - ignored
			DI => writeInput,			-- Input data port, width=WRITE_WIDTH
			WE => cw(3 downto 2),		-- RAM is byte read, high or low byte
			WRADDR => vecAddrWrite,		-- Write address, width=F(BRAM_SIZE)
			WRCLK => clk,				-- 1-bit input write clock
			WREN => cw(4));			-- 1-bit master write enable
end behavior;

